File System
1 Introduction

The purpose of this document is to show how to use the AsmL toolkit for model based analysis and testing of software. In section 2 we present an AsmL model of basic file system operations. Section 3 contains the simulation part, so the reader can run the model and observe random behaviors of the model. Then, in section 4 we use the model to generate test sequences for the file operations. Section 5 contains discussion about the quality of the generated test suite.
2 The file system model 

Before describing the file system, we introduce basic types for file attributes such as file identifier, file name, file time etc. The model described below covers only very common file system characteristics. We ignore volumes, references, access restrictions and so on, although the model could easily be extended in the corresponding directions.

type FileId   = Integer // file identifier
type FileTime = Integer // time stamp assigned to a file
type FileName = String  // local file name including extension
enum FileType

  Directory
  Regular
2.1 File attributes
All the file parameters are gathered in the FileInfo class. Fields fid and sort are constants, the rest are variable values.
class FileInfo // model of a file
  fid  as FileId                  // file identifier
  sort as FileType                // directory or regular 
  var name     as FileName        // file name relative to the parent dir
  var parent   as FileId          // identifier of the parent directory
  var data     as String          // data piled up in the file (ignored here)
  var children as Set of FileId   // set of identfiers of children resourses
  var creationTime   as FileTime  // creation time stamp
  var lastUpdateTime as FileTime  // time stamp of last update 
2.2 File system constructor 
The only fields of the file system class are: reference to the root directory root, information about files, and timer. 
class FileSystem

  root as FileId               // (= 0) file identifier of the root directory
  var files as Set of FileInfo // files kept by the file system
  var now   as FileTime        // current time
  FileSystem() // constructor of the file system class
    root  = 0
    files = { new FileInfo(root, Directory, ".", 0, "", {}, 0, 0) }

    now   = 1
2.3 Auxiliary functions  

We need the function File to find a file by its fid value. The function Known checks if fileId value is already used for some file. The function Now returns current timer value and then increments the timer.
class FileSystem

  File(fid as FileId) as FileInfo // search file record by FileId vaule
    require Known(fid)
    return the file | file in files where file.fid = fid

  Known(fid as FileId) as Boolean // 
    return fid in { file.fid | file in files } // the value is used
                                               // for an exisitng file
  Now() as FileTime  // current time
    now += 1
    return now

2.4 File operations
The main actions of our model correspond to the standard file operations: creation, update, moving/renaming, and deletion of a file.

2.4.1 Create file

To create file we have to check first that the parent exists in the file system, and that the file name does not clash with an existing file under the same parent.
class FileSystem 
  CreateEnabled(      // preconditions for creation of a file
    parent as FileId,

    name as String

    ) as Boolean

    return Known(parent)

      and then (File(parent).sort = Directory)

      and then name notin { File(ch).name | ch in File(parent).children }

When file is created we add the corresponding record to files and update the parent record accordingly.
class FileSystem  
  Create(

    parent as FileId,

    name as String,

    data as String,

    sort as FileType

    ) as FileId
    require CreateEnabled(parent, name)

    let fid = min id | id in {1..1000} where not Known(id)

    let time = Now()

    add new FileInfo(fid, sort, name, parent, data, {}, time, time) to files
    File(parent).children += { fid }

    return fid
2.4.2 Update file    

The only precondition for the Update action is that the file should exists.
class FileSystem

  UpdateEnabled(fid as FileId) as Boolean

    return Known(fid)
The transition consists in updating the data and lastUpdateTime fields. 
class FileSystem

  Update(

    fid as FileId,

    data as String

    )

    require UpdateEnabled(fid)

    File(fid).data := data
    File(fid).lastUpdateTime := Now()
2.4.3 Delete file
To delete file we need to ensure first that the file exists; it is not the root, and the set of children is empty.
class FileSystem

  DeleteEnabled(fid as FileId ) as Boolean    
    return Known(fid)
           and then fid ne root
           and then File(fid).children = {}

Then we remove the file from files and update the parent record in the correspondingly.
class FileSystem

  Delete(fid as FileId)

    require DeleteEnabled(fid)

    let parent = File(File(fid).parent)
    remove fid from parent.children
    remove File(fid) from files

2.4.4 Move/rename file    
The moving/renaming operation has a several preconditions – all of them are clear.
class FileSystem

  MoveEnabled(

    fid as FileId,

    newParent as FileId,

    newName   as String    
    ) as Boolean  
    return Known(fid) and Known(newParent) 

           and then File(newParent).sort = Directory

           and newParent notin Descendants(fid)

           and newName notin { File(ch).name | ch in File(newParent).children}    

  Move(

    fid as FileId,

    newParent as FileId,

    newName   as String

    )
    require MoveEnabled(fid, newParent, newName)

    File(fid).name := newName    
    let oldParent = File(fid).parent

    if oldParent ne newParent then    
      remove fid from File(oldParent).children

      add fid to File(newParent).children

      File(fid).parent := newParent
2.4.5 Auxiliary functions

These are needed to define precondition for the Move transition.
class FileSystem

  Descendants( depth as Integer, fid   as FileId ) as Set of FileId

    if depth<0 then 

      return {}

    else
      return { fid } union 

               BigUnion({ File(ch).children | ch in Descendants(depth-1,fid) })

  Descendants( fid as FileId ) as Set of Integer

    return Descendants(Size(files), fid)
3 Random scenario execution

3.1 Random action

class FileSystem

  RandomCreate()

    choose parent in { file.fid | file in files },

           name in { "a", "b", "c" },

           data in { "data1", "data2" },

           sort in { Regular, Directory }

      where CreateEnabled(parent, name)

      Create(parent, name, data, sort)

  RandomUpdate()

    choose fid in { file.fid | file in files },
           data in { "data" + ToString(i) | i in { 1..100 } }

      where UpdateEnabled(fid)

      Update(fid, data)

  RandomMove()

    choose fid in { file.fid | file in files },
           newParent in { file.fid | file in files },
           newName in { "A", "B", "C" }

      where MoveEnabled(fid, newParent, newName)

      Move(fid, newParent, newName)

  RandomDelete()

    choose fid in { file.fid | file in files } 

      where DeleteEnabled(fid)

      Delete(fid)

3.2 Monitoring

class FileSystem

  ShowState()

    step forall f in files where f.fid ne root
      let pName = File(f.parent).name
      WriteLine( ToString( (f.fid, pName + "/" + f.name, f.parent, f.data) ) )

    WriteLine("")

3.3 The main loop
var fs = new FileSystem()

Main()

  let fs = new FileSystem()
  let actions = { "Create", "Update", "Move/Rename", "Delete" }
  step
    fs.RandomCreate()
  step 

    fs.ShowState()
  step for i = 1 to 10
    step choose action in actions

      match action 

        "Create"      : fs.RandomCreate()

        "Update"      : fs.RandomUpdate()

        "Move/Rename" : fs.RandomMove()

        "Delete"      : fs.RandomDelete()

      WriteLine( "Step " + ToString(i) + ": Random " + action)

    step 

      fs.ShowState()
Sorted(list as Seq of Object) as Seq of String
  var res as Seq of String = [ ToString (el) | el in list ]

  step until fixpoint
    choose i in [0..Size(res)-2],

           j in [i+1..Size(res)-1] 

      where i<j and res(i)>res(j)

      res(i) := res(j)

      res(j) := res(i)

  step
    return res
4 Generating test sequences
We use the following property to define equivalence classes on the state space.
class FileSystem

  Structure() as Set of (FileId, FileId)

    return { (f.parent,f.fid) | f in files where f.fid ne 0 }

In addition, we stipulate that size of the file system should be less then 4 (root plus at most 3 files). The rest of the configuration is the same as for the random scenario execution. The resulting finite state machine (FSM) is shown on the picture (it took approximately 3 minutes to generate it). Then, the tool allows us to produce a test suite covering all the links in this graph.
[image: image1.wmf]
FSM produced by the AsmL Test Tool 1
The test sequences are saved in XML, so one can use them as input for testing tools.
5 Discussion
Questions

1. What are good strategies for approximating FSM? (Currently the tool has only 2 algorithms implemented.)

2. Is the produced FSM complete? How can we be sure that no links are overlooked?
3. Given the produced FSM, what are reasonable methods for making test sequences? We can either cover all links in the graph, or all cycles, or all chains of a fixed length etc.
4. What properties are reasonable to choose for defining the equivalence classes?






